Satellite

In this section we apply **homalg** and **OreModules** to a linear system describing a satellite in a circular equatorial orbit. See [Kai80, p. 60 and p. 145] and [Mou95, p. 6 and p. 11] and the *Library of Examples* at [CQR07]. We load the package **homalg** and the ring-specific package **OreModules** providing procedures for the algebraic analysis of linear systems over Ore algebras.

```
> restart;
```

```
> with(OreModules):
```

```
> with(homalg):
```

Since we only use the ring-specific package OreModules, we set the default package for homalg to OreModules:

```
> `homalg/default`:=`OreModules/homalg`;
```

$$homalg/default := OreModules/homalg$$

We define the Weyl algebra $Alg = A_1$, where Dt acts as differentiation w.r.t. time t. Note that we have to declare the parameters ω (angular velocity), m (mass of the satellite), r (radius component in the polar coordinates), a and b of the system in the definition of the Ore algebra:

```
> Alg:=DefineOreAlgebra(diff=[Dt,t], polynom=[t],
comm=[omega,m,r,a,b]):
```

The linearized ordinary differential equations for the satellite in a circular orbit are given by the following matrix R. These equations describe the motion of the satellite in the equatorial plane, where the fifth and the sixth column of R incorporate the controls $u1$, $u2$ which represent radial thrust resp. tangential thrust caused by rocket engines (see [Kai80, p. 60 and p. 145]).

```
> R:=matrix([[Dt,-1,0,0,0,0], [-3*omega^2,Dt,0,-2*omega*r,-a/m,0],
[0,0,Dt,-1,0,0], [0,2*omega/r,0,Dt,0,-b/(m*r)]]);
```

$$R := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\, r & -\frac{a}{m} & 0 \\ 0 & 0 & Dt & -1 & 0 & 0 \\ 0 & \frac{2\,\omega}{r} & 0 & Dt & 0 & -\frac{b}{m\, r} \end{bmatrix}$$

We find a presentation of the module associated with the linear system over the Weyl algebra A_1, i.e. of the cokernel of R:

```
> M:=Cokernel(R, Alg);
```

$$M := [[[1, 0, 0] = [0, 0, 1, 0, 0, 0], [0, 1, 0] = [0, 0, 0, 0, 1, 0], [0, 0, 1] = [0, 0, 0, 0, 0, 1]], [[Dt^2\, m\, r\, \omega^2 + Dt^4\, m\, r,\ 2\,\omega\, Dt\, a,\ -Dt^2\, b + 3\,\omega^2\, b]], \text{“Presentation”}, -\frac{s^3+s^2+s+1}{-1+s} + \frac{2}{(-1+s)^2}]$$

We compute the formal adjoint of the differential operator R:

```
> R_adj:=Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \frac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\, r & -1 & -Dt \\ 0 & -\frac{a}{m} & 0 & 0 \\ 0 & 0 & 0 & -\frac{b}{m\, r} \end{bmatrix}$$

Some structural properties of the linear system under consideration are determined by computing the extension modules with values in Alg of the cokernel of R_{adj}. We compute the first extension module:

```
> Ext_R(1, R_adj, Alg);
```

$$\left[\left[1 = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}\right], [1], \text{“Presentation”}, 0\right]$$

From this presentation we see that the first extension module is zero. Therefore, the torsion submodule of the cokernel of R is zero. Hence, the system of the satellite is controllable.

```
> TorsionSubmodule(R, Alg);
```

$$[[1 = [0, 0, 0, 0, 0, 0]], [1], \text{“Presentation”}, 0]$$

The next three statements demonstrate that this torsion submodule was computed by `homalg` using the procedure `ParametrizeModule` which returns a differential operator P such that the composition of R and P is zero. P defines a parametrization of the linear system given by R if and only if the kernel of $(.P)$ equals the image of $(.R)$, which means that the complex defined by these morphisms is exact. If we consider functions in an injective cogenerator (e.g. smooth functions, [CQR05, Zer00]), then we have $Ry = 0$ if and only if $y = P\xi$ for some vector of functions ξ. In general, P defines an embedding of the biggest possible factor module of the cokernel of R into a free module.

```
> P:=ParametrizeModule(R, Alg);
```

$$P := \begin{bmatrix} 0 & b\,a \\ 0 & b\,a\,Dt \\ b\,a & 0 \\ b\,a\,Dt & 0 \\ -2\,Dt\,b\,\omega\,r\,m & -3\,b\,m\,\omega^2 + Dt^2\,b\,m \\ a\,Dt^2\,m\,r & 2\,a\,Dt\,m\,\omega \end{bmatrix}$$

```
> Compose(R, P, Alg);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix}$$

```
> DefectOfHoms(R, P, Alg);
```

$$[[1 = [0, 0, 0, 0, 0, 0]], [1], \text{“Presentation”}, 0]$$

Since the system is controllable, we now check whether the system is flat [FLMR95, CQR05]. Every left-inverse of the parametrization P gives a flat output of the system:

```
> S:=Leftinverse(P, Alg);
```

$$S := \begin{bmatrix} 0 & 0 & \frac{1}{b\,a} & 0 & 0 & 0 \\ \frac{1}{b\,a} & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Therefore, $(\xi 1 : \xi 2)^T = S(x1 : x2 : x3 : x4 : u1 : u2)^T$ is a flat output of the system which satisfies $(x1 : x2 : x3 : x4 : u1 : u2)^T = P(\xi 1 : \xi 2)^T$. We notice that this flat output exists only if $ab \neq 0$. Hence, in the generic case the system is flat. Equivalently, the cokernel of R is free and, in particular, projective. Let us remember that the full row-rank matrix R admits a right-inverse if and only if the cokernel of R is projective. By the theorem of Quillen-Suslin, for modules over commutative polynomial rings, projectiveness is the same as freeness. So, M is projective which we could also have discovered by succeeding to compute a right-inverse of R:

```
> Rightinverse(R, Alg);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ -\frac{Dt\, m}{a} & -\frac{m}{a} & \frac{2\,\omega\, r\, m}{a} & 0 \\ -\frac{2\,\omega\, m}{b} & 0 & -\frac{Dt\, m\, r}{b} & -\frac{m\, r}{b} \end{bmatrix}$$

Following [Mou95], we modify the description of the control of the satellite in the system. If the rocket engines are commanded from the earth, then, due to transmission time, a constant time-delay occurs in the system. Hence, we enlarge the above Ore algebra by a shift operator δ:

```
>  Alg2:=DefineOreAlgebra(diff=[Dt,t], dual_shift=[delta,s],
polynom=[t,s], comm=[omega,m,r,a,b], shift_action=[delta,t]):
```

The system matrix is given as follows:

```
>  R2:=matrix([[Dt,-1,0,0,0,0],
[-3*omega^2,Dt,0,-2*omega*r,-a*delta/m,0], [0,0,Dt,-1,0,0],
[0,2*omega/r,0,Dt,0,-b*delta/(m*r)]]);
```

$$R2 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\, r & -\frac{a\,\delta}{m} & 0 \\ 0 & 0 & Dt & -1 & 0 & 0 \\ 0 & \frac{2\,\omega}{r} & 0 & Dt & 0 & -\frac{b\,\delta}{m\, r} \end{bmatrix}$$

We define a formal adjoint $R2_{adj}$ of $R2$ using an involution of $Alg2$:

```
>  R2_adj:=Involution(R2, Alg2);
```

$$R2\_adj := \begin{bmatrix} -Dt & -3\,\omega^2 & 0 & 0 \\ -1 & -Dt & 0 & \frac{2\,\omega}{r} \\ 0 & 0 & -Dt & 0 \\ 0 & -2\,\omega\, r & -1 & -Dt \\ 0 & \frac{a\,\delta}{m} & 0 & 0 \\ 0 & 0 & 0 & \frac{b\,\delta}{m\, r} \end{bmatrix}$$

We check controllability and parametrizability of the system:

```
>  Ext_R(1, R2_adj, Alg2);
```

$$\left[\left[1 = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \right], [1], \text{“Presentation”}, 0 \right]$$

We find that the first extension module with values in $Alg2$ of the cokernel of $R2_{adj}$ is genericall y the zero module. Equivalently, the system is generically controllable i.e. parametrizable.We continue to study the structural properties of the system by examining the algebraic properties of the cokernel of $R2$. The next step is to compute the second extension module with values in $Alg2$ of N:

```
>  Ext_R(2, R2_adj, Alg2);
```

$$[[[1,\,0] = \begin{bmatrix} 0 \\ 1 \end{bmatrix},\, [0,\,1] = \begin{bmatrix} 1 \\ 0 \end{bmatrix}],\, [[0,\,\delta],\, [\delta,\,0],\, [-2\,\omega\,Dt,\, Dt^2 - 3\,\omega^2],\, [Dt^2,\, 2\,\omega\,Dt]],$$
$$\text{“Presentation”},\, \frac{2\,(s+1)}{(-1+s)^2}]$$

The second extension module is not zero. Hence, the cokernel of $R2$ is not projective. Since $R2$ has full row-rank, this is equivalent to the fact that $R2$ does not admit a right-inverse:

```
> Rightinverse(R2, Alg2);
```

$$FAIL$$

In the special case where $a = 1$ and $b = 0$, we have the following system matrix:

```
> R20:=subs(a=1,b=0, copy(R2));
```

$$R20 := \begin{bmatrix} Dt & -1 & 0 & 0 & 0 & 0 \\ -3\,\omega^2 & Dt & 0 & -2\,\omega\,r & -\dfrac{\delta}{m} & 0 \\ 0 & 0 & Dt & -1 & 0 & 0 \\ 0 & \dfrac{2\,\omega}{r} & 0 & Dt & 0 & 0 \end{bmatrix}$$

The a presentation of the first extension module with values in $Alg2$ of the cokernel of the formal adjoint of $R20$ is given by:

```
> Ext_R(1, Involution(R20, Alg2), Alg2);
```

$$\left[\left[1 = \begin{bmatrix} 2\,\omega\,r \\ 0 \\ 0 \\ 0 \\ 0 \\ 4\,\omega^2 \end{bmatrix}\right],\, [Dt],\, \text{“Presentation”}\,,\, -\frac{1}{(-1+s)^3}\right]$$

Hence, we find a torsion element of the cokernel of $R20$ which corresponds to an autonomous element of the satellite system. Using the procedure `TorsionSubmodule` of **homalg** this presentation can be obtained directly:

```
> TorsionSubmodule(R20, Alg2);
```

$$[[1 = [6\,m\,\omega^2,\, 0,\, 0,\, 3\,\omega\,r\,m,\, 0,\, 0]],\, [Dt],\, \text{“Presentation”},\, -\frac{1}{(-1+s)^3}]$$

Author: Daniel Robertz

Date:	2005-12-31
Last modified:	2007-06-22 17:41

References

[BR] Mohamed Barakat and Daniel Robertz, **homalg** – *A meta-package for homological algebra*, submitted. arXiv:math.AC/0701146 and (http://wwwb.math.rwth-aachen.de/homalg).

[BR07] ———, **homalg** *project*, 2004-2007, (http://wwwb.math.rwth-aachen.de/homalg).

[CQR05] F. Chyzak, A. Quadrat, and D. Robertz, *Effective algorithms for parametrizing linear control systems over Ore algebras*, Appl. Algebra Engrg. Comm. Comput. **16** (2005), no. 5, 319–376, (http://www-sop.inria.fr/cafe/Alban.Quadrat/PubsTemporaire/AAECC.pdf). MR MR2233761 (2007c:93041) 2

[CQR07] Frédéric Chyzak, Alban Quadrat, and Daniel Robertz, *OreModules: a symbolic package for the study of multidimensional linear systems*, Applications of time delay systems, Lecture Notes in Control and Inform. Sci., vol. 352, Springer, Berlin, 2007, (http://wwwb.math.rwth-aachen.de/OreModules), pp. 233–264. MR MR2309473 1

[FLMR95] Michel Fliess, Jean Lévine, Philippe Martin, and Pierre Rouchon, *Flatness and defect of non-linear systems: introductory theory and examples*, Internat. J. Control **61** (1995), no. 6, 1327–1361. MR MR1613557 2

[Kai80] Thomas Kailath, *Linear systems*, Prentice-Hall Inc., Englewood Cliffs, N.J., 1980, Prentice-Hall Information and System Sciences Series. MR MR569473 (82a:93001) 1

[Mou95] H. Mounier, *Propriétés des systèmes linéaires à retards: aspects théoriques et pratiques*, Ph.D. thesis, University of Orsay, France, 1995. 1, 3

[Zer00] Eva Zerz, *Topics in multidimensional linear systems theory*, Lecture Notes in Control and Information Sciences, vol. 256, Springer-Verlag London Ltd., London, 2000. MR MR1781175 (2001e:93002) 2

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: daniel@momo.math.rwth-aachen.de